An Introduction to Dynare[1]

by Michael Siemer (Boston University)

at Studienzentrum Gerzensee, Switzerland
Doctoral Course

February, 2010

[1] This presentation borrows heavily from notes by M. Julliard

What is Dynare?

- DYNARE: A software for the **simulation** and **estimation** of rational expectation models
- developed by researchers headed by M. Julliard and including Tommaso Mancini Griffoli (SZG)
- collection of functions (300+) for **Matlab** (other platforms also available), install routines for Windows, Mac OS, Linux (Debian, Ubuntu)
- For download at `www.dynare.org` free of charge
- After downloading and installation make sure to add the dynare folder to your matlab path

What is Dynare?

- DYNARE: A software for the **simulation** and **estimation** of rational expectation models
- developed by researchers headed by M. Julliard and including Tommaso Mancini Griffoli (SZG)
- collection of functions (300+) for **Matlab** (other platforms also available), install routines for Windows, Mac OS, Linux (Debian, Ubuntu)
- For download at `www.dynare.org` free of charge
  - Dynare User Guide at `http://www.dynare.org/documentation-and-support/user-guide`
  - Practicing Dynare at `http://homepages.nyu.edu/~ts43/research/AP_tom16.pdf`, written for an earlier version of Dynare, some examples might not work.

'Practicing Dynare'

Provides a number of examples for Dynare. Among them are

- Cagan (1956): Demand for money during hyperinflations
- Kim and Kim (2003): International Real Business Cycle Model with Complete Markets
- Bansal and Yaron (2004): Asset pricing with risks for the long-run

Outline

Features

Dynare

- solves for the steady state of DSGE model
- computes **first or second order approximation** of linear/nonlinear stochastic models[2].
  $\longrightarrow$ more generally the expression for these approaches is **perturbation method**
- estimates DSGE models using **Bayesian Maximum Likelihood**
- optimal policy under commitment (Ramsey policy)
- little programming skills required (that may be a lie as we will discover)

[2] also solves deterministic models but those have become somewhat rare

The general problem

$$E_t \left\{ f(y_{t+1} y_t, y_{t-1}, u_t; \theta \right\} = 0$$

- where $f(\cdot)$ - are functions - y_t is a vector of endogenous variables that contains both forward looking variables and predetermined variables, u_t is a vector of exogenous shocks, θ are exogenous parameters
- $E(u_t) = 0$, $E(u_t, u_t') = \Sigma_u$, $E(u_t, u_s') = 0$ for $s \neq t$.
- In a stochastic framework, the unknowns are the decision functions:

$$y_t = g(y_{t-1}, u_t) \tag{1}$$

First order Approximation

Equation (1) is approximated in the following way:

$$y_t = \overline{y} + A\hat{y}_{t-1} + Bu_t \tag{2}$$

where $\hat{y}_t = y_t - \overline{y}$ and $\overline{y}$ is steady state.

- A first order approximation is nothing else than a standard solution through linearization
- A first order approximation in terms of the logarithm of the variables provides standard log-linearization
- Dynare uses a method proposed by Klein (2000) and Sims (2002).
- Alternative solution methods were developed e.g. King and Watson (2002), Anderson and Moore (1985)
  $\rightarrow$ one difference to KW, AM is that while you have to log-linearize the model and find the steady state yourself for those, Dynare can do it for you.
- Note: first order approximation is probably the most frequently used method

Second order Approximation

Equation (1) is approximated in the following way:

$$y_t = \overline{y} + A\hat{y}_{t-1} + Bu_t + \frac{1}{2}\left(\hat{y}'_{t-1}C\hat{y}_{t-1} + u'_t D u_t\right) + \hat{y}'_{t-1}Fu_t + G\Sigma_u \quad (3)$$

- Dynare uses a method proposed by Sims(2002), Schmitt-Grohe and Uribe (2003) and Collard and Julliard (2000)
- two features of second order:
  1. decision rules and transition functions are 2nd order polynomials
  2. departure from certainty equivalence: the variance of future shocks matters, particularly important for welfare calculation
- Note: second order approximation is being used more and more in economics

Standard procedure if NOT using Dynare

- find model equations (FOCs, other equilibrium conditions)
- find the steady state
- identify endogenous variables, predetermined and exogenous variables
- (log)-linearize the equations
- cast the equations in this kind of framework (this can vary slightly depending on the method you are using):

$$AE_t Y_{t+1} = BY_t + C_0 X_t + C_1 E_t X_{t+1} + \dots C_n E_t X_{t+n}$$

- specify exogenous process
- write model in state space form

Basics: Dynare .mod file

Can be written in Matlab and contains instructions for Dynare. Consist of four blocks

- **preamble:** lists variables and parameters
- **model:** equations of the model
- **steady state or initial value:** either advises Dynare to find steady state or provides the starting point for simulations or impulse response functions
- **shocks:** defines the shocks to the system
- **computation:** instructs Dynare to do certain operations: simulate the model, impulse response functions, estimate the model, etc.......

Dating convention

Dynare will automatically recognize predetermined and nonpredetermined variables, but you must observe a few rules:

- period t variables are set during period t on the basis of the state of the system at period t-1 and shocks observed at the beginning of period t.
- therefore, stock variables must be on an end-of-period basis: investment of period t determines the capital stock at the end of period t.
- Note: Dynare 4.1 (released a few weeks ago) allows you with a new command to use the more conventional notation.

Let's start with a simple example and create our own .mod file

Setup

A representative's household problem is[3]

$$\max_{\{c_t, l_t\}_{t=0}0^\infty} E_0 \sum_{t=0}^{\infty} \beta^t \frac{\left(c_t^\theta (1-l_t)^{1-\theta}\right)^{1-\gamma}}{1-\gamma}$$

subject to the resource constraint

$$c_t + i_t = e^{z_t} k_t^\alpha l_t^{1-\alpha} \quad (4)$$

the law of motion for capital

$$k_{t+1} = i_t + (1-\delta) k_t \quad (5)$$

and the stochastic process for productivity with $\epsilon_t \sim N(0,1)$

$$z_t = \rho z_{t-1} + \sigma \epsilon_t \quad (6)$$

[3] this example is taken from 'Practicing Dynare' mentioned earlier

Equilibrium conditions

Combine (4) and (5) to get

$$k_{t+1} = e^{z_t} k_t^{\alpha-1} l_t^{\alpha} - c_t + (1-\delta) k_t \tag{7}$$

The Euler equation

$$\frac{\left(c_t^{\theta}(1-l_t)^{1-\theta}\right)^{1-\gamma}}{c_t} = \beta E_t \left[\frac{\left(c_t^{\theta}(1-l_t)^{1-\theta}\right)^{1-\gamma}}{c_{t+1}} (1 + \alpha e^{z_{t+1}} k_{t+1}^{\alpha-1} l_{t+1}^{\alpha} - \delta) \right] \tag{8}$$

consumption/leisure choice

$$\frac{1-\theta}{\theta} \frac{c_t}{1-l_t} = (1-\alpha) e^{z_t} k_t^{\alpha-1} l_t^{\alpha} \tag{9}$$

The equilibrium is then characterized by (6), (7), (8), (9)

The .mod file for example I

```
//1. Preamble
var c k lab z;
varexo e;

parameters beta, theta, delta, alpha, gamma, rho, sigma;

beta = 0.987;
theta = 0.357;
delta = 0.012;
alpha = 0.4;
gamma = 2;
rho = 0.95;
sigma = 0.007;
```

The .mod file for example I

//1. Preamble

⋮

```
// 2. Model

model;
    (c^theta*(1-lab)^(1-theta))^(1-gamma)/c
    =beta*((c(+1)^theta*(1-lab(+1))^(1-theta))^(1-
gamma)/c(+1))*(1+alpha*exp(z(+1))*k^(alpha-1)*lab(+1)^(1-alpha)-delta);
    c=theta/(1-theta)*(1-alpha)*exp(z)*k(-1)^alpha*lab^(-alpha)*(1-lab);
    k=exp(z)*k(-1)^alpha*lab^(1-alpha)-c+(1-delta)*k(-1);
    z=rho*z(-1)+sigma*e;
end;
```

The .mod file for example I

```
//1. Preamble
      ⋮
//2. Model
      ⋮
// 3. Steady state or initial value

initval; //this is either the exact steady state or an approximation and
dynare then tries to find the true steady state.
k = 1;
c = 1;
lab = 0.3;
z = 0;
e = 0;
end;
```

The .mod file for example I

```
//1. Preamble
        ⋮
//2. Model
        ⋮
// 3. Steady state or initial value
        ⋮
// 4. Define the shocks

shocks;
var e;
stderr sigma;
end;
```

The .mod file for example I

```
// 1. Preamble
// 2. Model
// 3. Steady state or initial value
// 4. Define the shocks
// 5. Computation

steady; // finds the steady state

check; // provides the eigenvalues

stoch_simul(periods=1000, irf=40, order=1); // simulates the model

dynasave ('simudata.mat');//this is just saving the results
```

Now let's switch to Matlab and run this program and see what the IRF of an RBC model looks like.

Files produced by Dynare

Suppose our file is called example.mod. then you need to trype in matlab 'dynare example' and dynare subsequently produces a few .m files

- example.m: main Matlab script for the model
- example_static.m: static model
- example_dynamic.m: dynamic model

Remarks:

- there are many options you can choose for stoch_simul. Check 'Userguide' p.29/30.
- be aware of what Dynare's default settings are. For example: Moments are by default unfiltered. (and the HP filter option does not filter the simulated moments, only for theoretical moments)
- If some variables in the IRF do not return to their steady state, (a) check if you have enough periods in your IRF, (b) check if your model is stationary (!)
- If you want to have log-linearization rather than linearization, then you have to change the equations somewhat. If you had before $1/c$ you have to write now $1/exp(c)$ and the same for all other variables.
- Note that Dynare can have problems in finding the steady state depending on the model complexity and the initial values.

New in Dynare 4.1

Ability to use conventional notation,

- before:

```
var y, k, i;
...
model;
k = i + (1-delta)*k(-1);
...
end;
```

- new:

```
var y, k, i;
predetermined_variables k;
...
model;
k(+1) = i + (1-delta)*k;
...
end;
```

New in Dynare 4.1

- since higher order approximation (3rd, 4th) are now sometimes being used, Dynare 4.1 supports 3rd order approximation
- Possibility to use Anderson-Moore Algorithm to compute decision rules
- new additions to Dynare occur every few months....

Estimation of DSGE Model: General

- becoming increasingly popular in economics
- more computationally intensive than calibration that we did before
- very little documentation in textbook but two options are (both in SZG library)
    1. Structural Macroeconometrics by DeJong
    2. Methods for Applied Macroeconomic Research by Canova
- Maximum likelihood estimation and Bayesian MLE are quite related: Bayesian MLE contains an additional prior (additional information)
- A good and understandable example for MLE of DSGE models is Ireland, Peter (2004): 'Technology Shocks in the New Keynesian Model'. Data and code available on `http://www2.bc.edu/~irelandp/pro-grams.html`
- Example for Estimation in 'Userguide' does not run in Dyare 4.1.........

Bayesian Estimation: Idea

- Uncertainty and a priori knowledge about the model and it's parameters are described by prior probabilities
- Confrontation to the data leads to a revision of the probabilities (posterior probabilities)
- point estimates are obtained by minimizing a loss function (analogous to economic decision under uncertainty)
- testing and model comparison is done by comparing posterior probabilities

Estimation in Dynare

Dynare estimates the structural parameters of a model based on a linear approximation:

$$E_t \left\{ f(y_{t+1} y_t, y_{t-1}, u_t; \theta \right\} = 0$$

Estimation steps in Dynare:

1. Computes steady state
2. linearizes the model
3. solves the linearized model
4. computes the log-likelihood via Kalman filter
5. finds the maximum of the likelihood or posterior mode
6. simulates posterior distribution with metropolis algorithm
7. computes various statistics on the basis of the posterior distribution
8. computes values of unobserved variables
9. computes forecasts and confidence intervals

Example I again

$$\max_{\{c_t, l_t\}_{t=0}0^\infty} E_0 \sum_{t=0}^{\infty} \beta^t \frac{\left(c_t^\theta (1-l_t)^{1-\theta}\right)^{1-\gamma}}{1-\gamma}$$

subject to the resource constraint

$$c_t + i_t = e^{z_t} k_t^\alpha l_t^{1-\alpha} \tag{10}$$

the law of motion for capital

$$k_{t+1} = i_t + (1-\delta) k_t \tag{11}$$

and the stochastic process for productivity with $\epsilon_t \sim N(0, \sigma^2)$

$$z_t = \rho z_{t-1} + \epsilon_t \tag{12}$$

Now, suppose we want to estimate θ, γ, ρ and the standard deviation of ϵ.

the .mod file

```
var c k lab z;
varexo e;

parameters beta delta alpha rho theta gamma sigma;

beta = 0.987;
delta = 0.012;
alpha = 0.4;
sigma = 0.007;

model;
same as before
end;

varobs c;
```

the .mod file cont.

```
initval;
k = 1;
c = 1;
lab = 0.3;
z = 0;
e = 0;
end;
```

the .mod file continued

```
estimated_params;
stderr e, inv_gamma_pdf, 0.95,30;
rho, beta_pdf,0.93,0.02;
theta, normal_pdf,0.3,0.05;
gamma, normal_pdf,2.1,0.3;
end;

estimation(datafile=simudata,mh_replic=1000,mh_jscale=0.9,conf_sig=0.9,
nodiagnostic, bayesian_irf);
```

the .mod file: Some comments

- note: you need as many shocks as you have observables
- because in this estimation Dynare updates the steady state at each iteration, it is significantly faster to give Dynare an external steady state file, needs to have the name: example.mod, → example_steadystate.m
- for bayesian estimation we need 'priors' and their distributions, options in Dynare are normal, gamma, beta, inverse gamma and uniform distribution.
- Where do you get the priors from? Micro estimates, existing studies, ...
- Informative vs. uninformative prior
- estimation can take a LONG time (mh_replic determines the number of replications for the Metropolis Hastings algorithm, default is 20000)

Let's run this in Matlab

- Pros
  1. specify a DSGE in linear or non-linear form, no need to write out state space matrices
  2. simulation of the solution of the model to produce various statistics of interest: moments, IRF, forecasts, etc by virtually pressing a button
  3. estimation of the model's parameters using MLE or Bayesian methods
  4. useful checking tool
- Cons
  1. the documentation is still relatively poor with little explanation on which routine does what
  2. relatively difficult to check for errors/mistakes
  3. there are a number of unsupported solution methods e.g. Projection Methods
  4. estimation is still quite preliminary
  5. backward compatibility not guaranteed
  6. it does not write your dissertation (unfortunately)

Tomorrow

- Dynare case study: Bernanke, Gertler and Gilchrist (1999): Financial Accelerator
- how to use Dynare with a log-linear system of equations
- we will discuss the effects of financial frictions on the economy
- Why is the interesting? Financial Crisis 2008/09